Tartalom

A logikai és a funkcionális programozás összehasonlítása

Mi a közös a funkcionális (FP) és logikai (LP) nyelvekben?

1. A matematikai változó fogalma: egyetlen még ismeretlen értéket jelöl, nem mutábilis (nem változtatható)
2. Ciklusok helyett: rekurzió, vagy más eszközök (pl. listanézet)

Miben más az LP (ill. a Prolog) mint az FP megközelítés?

3. Az FP a lambda-kalkuluson alapul, az LP alapja az elsőrendű logika (FOL)
   - Az LP-ben függvények helyett predikátumokat (relációkat, vö. adatbáziskezelés) kell definiálnunk, predikátum $\equiv$ eljárás
   - Egy LP program futása 0, 1, vagy több eredményt adhat – nemdeterminisztikus keresés, vö. SQL lekérdezések
4. A Prolog szintaxisa kiterjeszthető – saját új operátorokat definiálhatunk
5. LP-ben a változók egy adatstruktúra (pl. lista) belsejében is előfordulhatnak – ezek a logikai változók pointerként működnek, pl. `[X,X,X]` egy olyan 3-elemű listát jelöl, amely csupa azonos (de egyelőre még ismeretlen) elemből áll
6. Az LP különösen jó szimbolikus feladatokra (pl. szimbolikus deriválás)
7. Sok FP nyelv van, de praktikusan a Prolog az egyetlen LP nyelv

Ismétlés: listák összefűzése

- A Prolog lista szintaxisa megegyezik az Elixir szintaxissal, de Prologban a változók kötelezően nagybetűvel vagy aláhúzásjellel (_) kezdődnek
- Idézzük fel a két listát összefűző `app` Elixir függvényt (`app/2`):

```
#   app(l1,   l2): l1 és l2 listák összefűzöttje (l1⊕l2)
def app([],    b) do              b end       # [] ⊕ b = b
def app([x|a], b) do [x|app(a,b)] end       # [x|a] ⊕ b = [x|a⊕b]
```

- Ennek egy 3-argumentumú Prolog predikátum felel meg, (`app/3`), itt a 3. argumentum lesz az összefűzött lista (az Elixir függvény eredménye):

```
%   app(L1, L2, L12): L1 és L2 listák összefűzöttje L12 (L1⊕L2=L12)
    app([],    B,              B).          % [] ⊕ B = B
    app([X|A], B,          [X|C]) :-        % [X|A] ⊕ B = [X|C] ha
           app(A, B, C).                    %      A ⊕ B = C
```

- A `C` segédváltozó az `A⊕B` részeredményt tárolja.
- Példa az `app/3` predikátum használatára:

```
| ?- app([1,2], [3,4], L).      ⟹    L = [1,2,3,4] ? ;
                                      no
```

- A fenti `app/3` eljárás `append` néven beépített predikátumként is elérhető.

Be- és kimenő argumentumok

- Az `app/3` predikátum az `app/2` Elixir függvény átírásával állt elő
- A Prolog predikátum azonban használható más módon is, pl:

```
| ?- app(L1, L2, [1,2]).
L1 = [], L2 = [1,2] ? ;
L1 = [1],  L2 = [2] ? ;
L1 = [1,2], L2 = [] ? ; no
```

- Az ún. I/O mód jelölésrendszer a különböző módú hívások leírására:
  - `+`: bemenő (input) arg., a hívás pillanatában behelyettesített
  - `-`: kimenő (output) arg., a hívás pillanatában behelyettesítetlen
  - `?`: be- és kimenő arg., tetszőleges Prolog kifejezés lehet
- Példák az `app(L1, L2, L3)` hívás különböző módú hívásaira:
  - `(+,+,+)`: konkatenálás ellenőrzése, pl. `app([1], [2], [1,2])` ⇒ `yes`
  - `(+,+,-)`: konkatenálás, pl. `app([1], [2], L3)` ⇒ `L3 = [1,2]`
  - `(+,-,+)`: két lista „különbsége”, pl. `app([1], L2, [1,2])` ⇒ `L2 = [2]`
  - `(+,-,-)`: nyílt végű lista előállítása, pl. `app([1], L2, L3)` ⇒ `L3 = [1|L2]`
  - `(-,-,+)`: lista szétszedése, pl. `app(L1, L2, [1,2])` ⇒ lásd fent
  - `(-,?,-)`: ∞ keresési tér: pl. `app(L1, [1], L3)` ⇒ `L1 = [], L3 = [1]? ;`

    `L1 = [A], L3 = [A,1]? ; L1 = [A,B], L3 = [A,B,1]? ;...`
- Az eredményekben logikai változók is lehetnek, lásd fenn, pl. `L2`, `A`, `B` stb.

Tartalom

A Prolog alapelemei: a családi kapcsolatok példája

- Adottak személyekre vonatkozó, adatbázis-szerű állítások, pl.

„gyerek–szülő” tábla

gyerek	szülő
Imre	István
Imre	Gizella
István	Géza
István	Sarolt
Gizella	C. Henrik
Gizella	B. Gizella

„férfiak” tábla

férfi
Imre
István
Géza
C. Henrik

- Rövidítések feloldása: C. Henrik $\Rightarrow$ Civakodó Henrik,
  B. Gizella $\Rightarrow$ Burgundi Gizella
- Definiáljuk az unoka–nagyszülő kapcsolatot, azaz hozzunk létre egy származtatott (virtuális) „unoka–nagyszülő” táblát!

A nagyszülő feladat — Prolog megoldás

- Egy Prolog program állításokból, ún. klózokból (clause) áll
- A legegyszerűbb klóz a tényállítás (fact), formája:

  `relációnév(`$\mathtt{Arg}_1$`, ..., `$\mathtt{Arg}_n$`).` (ez egy `klózfej`)
- A `relációnév` egy névkonstans (atom): kisbetűvel kezdődő azonosító vagy aposztrófok közé zárt tetsz. karaktersorozat (első közelítésben)
- Az argumentumok lehetnek névkonstansok, változók, stb.
- A változókat nagybetűvel kezdődő azonosítókkal – pl. `Gy`, `Sz` – jelöljük
- Az Imre herceg őseit leíró adatbázis-táblák Prolog alakja:

```
% sz(Gy, Sz): Gy szülője Sz.
  sz('Imre',    'Gizella').    % (sz1)
  sz('Imre',    'István').     % (sz2)
  sz('István',  'Sarolt').     % (sz3)
  sz('István',  'Géza').       % (sz4)
  sz('Gizella', 'B. Gizella'). % (sz5)
  sz('Gizella', 'C. Henrik').  % (sz6)
```

```
% ffi(Személy): Személy férfi.
  ffi('Imre').       % (f1)
  ffi('István').     % (f2)
  ffi('Géza').       % (f3)
  ffi('C. Henrik'). % (f4)
```

- A predikátumok jelentését egy `% fejkomment`-tel írjuk le, `/* ez is komment */`
- Azonos nevű és argumentumszámú klózok sorozata egy predikátumot alkot, pl. a fenti klózok az `sz/2` ill. `ffi/1` predikátumokat

A nagyszülő feladat — Prolog megoldás (folyt.)

- A klózok másik fajtája az ún. szabály (rule), formája:

```
klózfej :- cél_1, ..., cél_k.        % klózfej ← cél_1 ∧ ... ∧ cél_k
        % ^--klóztörzs--^
```

- A `cél` (goal), más néven hívás (call) szintaxisa (azonos a `klózfej`-évél):

```
relációnév(Arg_1, ..., Arg_n)
```

- A „nagyszülője” kapcsolatot definiáló szabály:

```
% Gyerek nagyszülője Nagyszulo.
nsz(Gyerek, Nagyszulo) :-        % Gyerek nagyszülője Nagyszulo ha ∃ Szulo
      sz(Gyerek, Szulo),               % Gyerek szülője Szulo és
      sz(Szulo, Nagyszulo).            % Szulo szülője Nagyszulo        (nsz)
```

- Egy program futtatásához egy célsorozatot (lekérdezést) kell megadni:

```
%    Ki Imre nagyapja? (pontosabban  Ki Imre férfi nagyszülője?)
| ?- nsz('Imre', NA), ffi(NA).       NA = 'C. Henrik' ? ;
                                     NA = 'Géza' ? ; no
%    Ki Géza unokája?
| ?- nsz(U, 'Géza').                 U = 'Imre' ? ; no
%    Ki Imre nagyszülője?
| ?- nsz('Imre', NSz).               NSz = 'B. Gizella' ? ;
                                     NSz = 'C. Henrik' ? ;
                                     NSz = 'Sarolt' ? ; NSz = 'Géza' ? ; no
```

Deklaratív szemantika – klózok logikai alakja

- A szabály jelentése egy implikáció: a törzsbeli célok konjunkciójából következik a fej.
  - Példa: `nsz(Gy,NSz) :- sz(Gy,Sz), sz(Sz,NSz).`
  - Logikai alak: $\forall \texttt{Gy}, \texttt{NSz}, \texttt{Sz}(\texttt{nsz}(\texttt{Gy}, \texttt{NSz}) \leftarrow \texttt{sz}(\texttt{Gy}, \texttt{Sz}) \wedge \texttt{sz}(\texttt{Sz}, \texttt{NSz}))$
  - Ekvivalens alak: $\forall \texttt{Gy}, \texttt{NSz}(\texttt{nsz}(\texttt{Gy}, \texttt{NSz}) \leftarrow \exists \texttt{Sz}(\texttt{sz}(\texttt{Gy}, \texttt{Sz}) \wedge \texttt{sz}(\texttt{Sz}, \texttt{NSz})))$
- A tényállítás feltétel nélküli állítás, pl.
  - Példa: `sz('Imre', 'István').`
  - Logikai alakja változatlan
  - Ebben is lehetnek változók, ezeket is univerzálisan kell kvantálni
- A célsorozat jelentése: keressük azokat a változó-behelyettesítéseket amelyek esetén a célok konjunkciója igaz
- Egy célsorozatra kapott válasz helyes, ha az adott behelyettesítésekkel a célsorozat következménye a program logikai alakjának – WHAT
- A Prolog garantálja a helyességet, de a teljességet nem: nem biztos, hogy minden megoldást megkapunk (kaphatunk hibajelzést, végtelen ciklust, végtelen keresési teret stb.) – HOW

Procedurális szemantika: az ún. redukciós modell

Redukciós lépés: egy CS_i célsorozat visszavezetése CS_{i+1}-re úgy, hogy

$$CS_i \leftarrow Program \wedge CS_{i+1}$$

Pl. az `(1)` célsorozat redukciója az `(nsz)` programklózzal `(2)`-t eredményezi:

```
:- nsz('Imre', NA),          ffi(NA).    (kiinduló célsorozat)   (1)
:- sz('Imre', Sz1), sz(Sz1, NA), ffi(NA).  (redukált célsorozat)   (2)
```

1. A klózt **lemásoljuk**, a változókat szisztematikusan újakra cserélve
   `nsz(Gy1, NSz1) :- sz(Gy1, Sz1), sz(Sz1, NSz1).` (nsz′)
2. `(1)`-et szétbontjuk, első cél: `nsz('Imre', NA)`, maradék célsor.: `ffi(NA)`.
3. Az első célt **egyesítjük** a klózfejjel, azaz változók behelyettesítésével a klózfejjel azonos alakra hozzuk (**kétirányú** mintaillesztés): behelyettesítés: `Gy1 = 'Imre', NSz1 = NA`, közös alak: `nsz('Imre', NA)`
4. Ha az egyesítés sikertelen, akkor a redukciós lépés meghiúsul, egyébként behelyettesítjük a klóztörzset: `sz('Imre', Sz1), sz(Sz1, NA)` és a maradék célsorozatot is (ebben most nincs változás): `ffi(NA)`
5. Új célsorozat = klóztörzs és utána a maradék célsorozat (lásd fenn, `(2)`)

Az `(1)`→`(2)` redukciós lépés értelmezhető az `(nsz)` „makró” kifejtéseként...

További redukciós lépések

A (2) célsorozat redukciója az (sz1) programklózzal:

```
:- sz('Imre', Sz1), sz(Sz1, NA), ffi(NA).                    (2)
```

1. Az (sz1) klóz nem tartalmaz változót, így nem szükséges lemásolni:

   ```
   sz('Imre', 'Gizella') /* :- (üres klóztörzs) */.          (sz1)
   ```

2. (2) első célja: `sz('Imre', Sz1)`, maradék célsor.: `sz(Sz1, NA), ffi(NA).`
3. Az első célt **egyesítjük** a klózfejjel behelyettesítés: `Sz1 = 'Gizella'`, közös alak: `sz('Imre', 'Gizella')`
4. A behelyettesített maradék: `sz('Gizella', NA), ffi(NA).`
5. Az új célsorozat: az (sz1) klóz (üres) törzse + a maradék célsorozat:

```
:- sz('Gizella', NA), ffi(NA).                               (3)
```

(Tényállítással redukálva 1-gyel csökken a célsorozat hossza!)

(3)-at redukálva (sz6)-tal (`sz('Gizella', 'C. Henrik').`) a `NA = 'C. Henrik'` behelyettesítést kapjuk, az új célsorozat:

```
:- ffi('C. Henrik').                                         (4)
```

(4)-et redukálva (f4)-gyel (`ffi('C. Henrik').`) üres célsorozatot (□) kapunk. Ezzel megállapítottuk, hogy az `NA = 'C. Henrik'` egy megoldás (1)-re.

A nagyszülő példa végrehajtása – egy teljes levezetés

```
% sz(Gy, Sz): Gy szülője Sz.
  sz('Imre',     'Gizella').    % (sz1)
  sz('Imre',     'István').     % (sz2)
  sz('István',   'Sarolt').     % (sz3)
  sz('István',   'Géza').       % (sz4)
  sz('Gizella',  'B. Gizella'). % (sz5)
  sz('Gizella',  'C. Henrik').  % (sz6)
```

```
% ffi(Személy): Személy férfi.
  ffi('Imre').        % (f1)
  ffi('István').      % (f2)
  ffi('Géza').        % (f3)
  ffi('C. Henrik').   % (f4)
```

```
% Gy nagyszülője NSz.
nsz(Gy, NSz) :-        sz(Gy, Sz), sz(Sz, NSz).          % (nsz)
```

```
(1)  :- nsz('Imre', NA), ffi(NA).                    (nsz): (1) ← (2)

(2)  :- sz('Imre', Sz1), sz(Sz1, NA), ffi(NA).       (sz1): (2) ← (3)
                                          Sz1 = 'Gizella'
(3)  :- sz('Gizella', NA), ffi(NA).                  (sz6): (3) ← (4)
                                          NA  = 'C. Henrik'
(4)  :- ffi('C. Henrik').                            (f4):  (4) ← (5)

(5)  □                                               (5) azonosan igaz
```

Bebizonyítottuk, hogy (1) teljesül az `NA = 'C. Henrik'` behelyettesítés esetén

A nagyszülő példa – a válasz követése az `answer` cél segítségével

```
% sz(Gy, Sz): Gy szülője Sz.
  sz('Imre',     'Gizella').    % (sz1)
  sz('Imre',     'István').     % (sz2)
  sz('István',   'Sarolt').     % (sz3)
  sz('István',   'Géza').       % (sz4)
  sz('Gizella',  'B. Gizella'). % (sz5)
  sz('Gizella',  'C. Henrik').  % (sz6)
```

```
% ffi(Személy): Személy férfi.
  ffi('Imre').        % (f1)
  ffi('István').      % (f2)
  ffi('Géza').        % (f3)
  ffi('C. Henrik').   % (f4)
```

```
% Gy nagyszülője NSz.
nsz(Gy, NSz) :-        sz(Gy, Sz), sz(Sz, NSz).                    % (nsz)
```

```
(1)  :- nsz('Imre', NA), ffi(NA),                 answer(NA).      (nsz): (1) ← (2)

(2)  :- sz('Imre', Sz1), sz(Sz1, NA), ffi(NA), answer(NA).      (sz1): (2) ← (3)

(3)  :- sz('Gizella', NA), ffi(NA),               answer(NA).      (sz6): (3) ← (4)

(4)  :- ffi('C. Henrik'),                  answer('C. Henrik').    (f4):  (4) ← (5)

(5)  :-                                    answer('C. Henrik').    A lekérdezés sikeres
```

A futás végén az `answer` „virtuális” cél tartalmazza a választ.

A nagyszülő példa végrehajtása – keresési tér

- A Prolog minden lehetséges redukciót szisztematikusan végigpróbál,
- balról jobbra haladó mélységi keresés formájában.




```
sz('Imre',    'Gizella').    % (sz1)
sz('Imre',    'István').     % (sz2)
sz('István',  'Sarolt').     % (sz3)
sz('István',  'Géza').       % (sz4)
sz('Gizella', 'B. Gizella'). % (sz5)
sz('Gizella', 'C. Henrik').  % (sz6)
```

```
ffi('Imre').       % (f1)
ffi('István').     % (f2)
ffi('Géza').       % (f3)
ffi('C. Henrik').  % (f4)
```

```
nsz(Gy, NSz) :-
    sz(Gy, Sz),
    sz(Sz, NSz).   % (nsz)
```

A Prolog végrehajtási algoritmus – első közelítés

Egy célsorozat végrehajtása

1. Ha az első cél beépített eljárást (BIP) hív, végrehajtjuk a BIP-et.
2. Ha az első cél felhasználói eljárásra vonatkozik, akkor megkeressük az eljárás első (visszalépés után: következő) olyan klózát, amelynek feje egyesíthető a hívással, és végrehajtjuk a redukciót.
3. Ha a redukció sikeres (találunk egyesíthető fejű klózt), folytatjuk a végrehajtást 1.-től az új célsorozattal.
4. Ha a redukció meghiúsul, akkor visszalépés következik:
   - visszatérünk a legutolsó, felhasználói eljárással történt (sikeres) redukciós lépéshez,
   - annak *bemeneti* célsorozatát megpróbáljuk *újabb* klózzal redukálni – ugrás a 2. lépésre
     (Ennek meghiúsulása értelemszerűen újabb visszalépést okoz.)

A végrehajtás nem „intelligens”

- Pl. `:- nsz(Gy, 'Géza').` hatékonyabb lenne ha a klóz törzsét jobbról balra hajtanánk végre
- DE: így a végrehajtás a program írója számára átlátható; a Prolog nem tételbizonyító, hanem programozási nyelv (WHAT rather than HOW)

A nagyszülő példa „tömörített” változata

- Imre herceget és felmenőit kétbetűs atomokkal jelöljük:
  Imre ⇒ `im`, Gizella ⇒ `gi`, István ⇒ `is`, Sarolt ⇒ `sa`, Géza ⇒ `ge`,
  Burgundi Gizella ⇒ `bg`, Civakodó Henrik ⇒ `ch`.




```
sz(im, gi).      % (sz1)
sz(im, is).      % (sz2)
sz(is, sa).      % (sz3)
sz(is, ge).      % (sz4)
sz(gi, bg).      % (sz5)
sz(gi, ch).      % (sz6)

ffi(im).         % (f1)
ffi(is).         % (f2)
ffi(ge).         % (f3)
ffi(ch).         % (f4)
```

```
nsz(U, NSz) :-
    sz(U, Sz),
    sz(Sz, NSz). % (nsz)
```

A cél-redukciós modell alapfogalmai

- A végrehajtás bemenete:
  - egy Prolog program (klózok sorozata), pl. a „nagyszülő” program, és
  - egy célsorozat, pl. `:- nsz(im, Sz).`
    a megoldás meghatározása érdekében ezt egy utolsó,
    `answer(Megoldás)` fiktív céllal bővítjük ki, pl.

```
:- nsz(im, NSz), answer(NSz).        % Kik Imre nagyszülei?
:- sz(Gy, Sz),   answer(Gy-Sz).      % Mik a gyerek-szülő párok?
```

- Az `answer(...)` cél segítségével követhetjük a megoldás felépülését
- Ha a célsorozat már csak az `answer` célt tartalmazza, akkor eljutottunk egy megoldáshoz (ezt a szerepet korábban az üres célsorozat játszotta)
- Az `answer` csak egy elméleti eszköz, nem beépített elj., de definálhatjuk, így: `answer(M) :- write(M), nl, fail.`
- A végrehajtásnak többféle kimenetele lehetséges:
  - Hiba (kivétel, exception), pl. `:- Y = alma, X is Y+1.`
    (Ezzel most nem foglalkozunk részletesebben.)
  - Meghiúsulás (nincs megoldás), pl. `:- sz(ge, Sz), answer(Sz).`
  - Siker (1 vagy több megoldás), pl. `:- sz(im, Sz), answer(Sz).`
- A végrehajtási modell gyakorlása ⇒ `https://ait.plwin.dev/P1-1`

A redukciós végrehajtás alapfogalmai (folyt.)

- A végrehajtás által használt (imperatív!) adatstruktúrák:
  - a jelenlegi célsorozatot tartalmazó változó (Goal)
  - a választási pontokat (VP) tartalmazó verem (Choice point stack)
- Például a `nsz(im, NA), ffi(NA), answer(NA)` célsorozat végrehajtásakor az alábbi VP verem jön létre:

Choice point stack

ChPoint name	Clause list	Goal	
CHP2	[sz5,sz6]	(3)	`sz(gi,NA),ffi(NA),answer(NA).`
CHP1	[sz1,sz2]	(2)	`sz(im,Sz),sz(Sz,NA),ffi(NA),answer(NA).`

- A VP verem akkor mélyül, ha 2 vagy több klózzal lehet redukálni
  - a redukció előtt a veremre elmentjük a célsorozatot és a redukcióban használható klózok listáját, majd folytatjuk a végrehajtást
  - ennek meghiúsulása esetén
    - a verem tetején levő klózlistából elhagyjuk az első elemet,
    - a klózlistában most első klózzal folytatjuk a redukciót,
    - ezt megelőzően, ha egyelemű a klózlista, megszűntetjük a VP-t
  - ha meghiúsuláskor üres a VP-verem $\Rightarrow$ kimerítettük a keresési teret

A redukciós modell folyamatábrája



A kettős nyilak jelentése: ugrás a halványlila körben megadott lépésre (folytatás a megfelelő sötétlila körnél).

Megjegyzések a folyamatábrához

- Hétféle végrehajtási lépésünk van: **S1**–**S7**, ahol **S1** a kiindulási pont (de közbülső is), **S7** a végállapot.
- Az **S1** lépés alapvető feladata az elágaztatás az **S2**–**S6** lépések egyikére
  - ha `Goal` már csak az `answer(...)` elemet tartalmazza ⇒ **S6**;
  - ha az első cél beépített eljárást hív ⇒ **S5**;
  - egyébként az első cél felhasználói eljárást hív. Ekkor megvizsgáljuk (általában csak közelítően), hogy az eljárás mely klózainak fejére illeszthető az első cél, és ezek száma (n) szerint ⇒ **S2**, **S3** vagy **S4**.
- **S2** létrehoz egy VP-t, majd az első klózzal redukál (⇒ **S3**).
- **S3** meghiúsulhat, ha **S1**-ben n csak közelítés volt, ilyenkor ⇒ **S4**.
- **S4** a VP-ban eltárolt következő klózzal való redukcióra lép (⇒ **S3**), ha van ilyen; egyébként befejezi a végrehajtást (⇒ **S7**).
- **S5** az **S3** lépéssel analóg módon vagy ⇒ **S1**, vagy ⇒ **S4**.
- **S6**-ban a megoldás megjelenítése után visszalépéssel folytatjuk (⇒ **S4**, további megoldások keresése).

A Prolog adatfogalma, a Prolog kifejezés (term)

- konstans (az `atomic` beépített eljárásnak felel meg)
  - számkonstans (`number`) – egész/lebegőpontos, pl. `1`, `-2.3`, `3.0e10`
  - névkonstans (`atom`), pl. `'István'`, `szuloje`, `*`, `++`, `tree_sum`
  - egy C konstans funktora $C/0$
- összetett- vagy struktúra-kifejezés (`compound`)
  - ún. kanonikus alak: $\langle$ struktúranév $\rangle(\langle \text{arg}_1 \rangle, \ldots, \langle \text{arg}_n \rangle)$
    - a $\langle$ struktúranév $\rangle$ egy névkonstans, az $\langle \text{arg}_i \rangle$ argumentumok tetszőleges Prolog kifejezések
    - a kifejezés funktora: $\langle$ struktúranév $\rangle$/n
    - példák: `person(ian,smith,2003)`, `<(X,Y)`, `is(X, +(Y,1))`
  - szintaktikus „édesítőszerek”, pl. operátorok és listák:
    `X is Y+1` $\equiv$ `is(X, +(Y,1))`, ill. `[1,2,3|X]` $\equiv$ `.(1,.(2,.(3,X)))`
- változó (`var`), pl. `Valtozo`, `X`, `_var`, `_` (don’t care) (nincs funktora)
  - a változó alaphelyzetben behelyettesítetlen, értékkel nem bír, egyesítés során egy tetszőleges Prolog kifejezést (akár egy másik változót) vehet fel értékül – dinamikus típusfogalom
  - a változó „first class citizen”, előfordulhat egy struktúra argumentumaként – logikai változó minta $\equiv$ adat

Néhány alapvető beépített eljárás (Built-In-Procedure, BIP)

- Kifejezések egyesítése
  - `X = Y`: az `X` és `Y` szimbolikus kifejezések egyesítése ≡ azonos alakra hozása változók esetleges behelyettesítésével, a lehető legáltalánosabb módon
  - `X \= Y`: az `X` és `Y` kifejezések nem egyesíthetőek (nem hozhatók azonos alakra)

```
| ?- U+V = 1+(2*3).   ⟹   U = 1, V = 2*3
| ?- U-V = (8-4)-2.   ⟹   U = 8-4, V = 2
| ?- U+1 = 4+V.       ⟹   U = 4, V = 1 ?      % Kétirányú a mintaillesztés!
| ?- U+1 \= V+4.      ⟹   yes % szimbolikus kifejezések, a + nem kommutatív!
```

- Típusvizsgálat (az elemi, nem bontható típusok aláhúzással jelölve)
  - `var(X)`: `X` változó, ill. ennek komplementere: `nonvar(X)`: `X` nem változó
  - `atomic(X)`: `X` konstans
    - `number(X)`: `X` szám (`float(X)`: `X` lebegőp., `integer(X)`: `X` egész)
    - `atom(X)`: `X` névkonstans
  - `compound(X)`: `X` összetett kifejezés

```
| ?- X = 1, atomic(X), number(X), integer(X).  ⟹   yes
| ?- atomic(X), X = 1.                         ⟹   no   (What rather than How)
```

A Prolog lista-fogalma

- A Prolog lista
  - Az üres lista a `[]` névkonstans.
  - A nem-üres lista a `'.'(Fej,Farok)` struktúra:
    - `Fej` a lista feje (első eleme), míg
    - `Farok` a lista farka, azaz a fennmaradó elemekből álló lista.
  - A listákat egyszerűsítve is leírhatjuk („szintaktikus édesítés”).
  - Megvalósításuk optimalizált, időben és helyben is hatékonyabb.
- A listák fastruktúra alakja és megvalósítása





- Az SWI Prolog nem szabványos, a lista-konstruktor nem `'.'`, hanem `'[|]'` :-(((

Listák jelölése – szintaktikus „édesítőszerek”

- Az alapvető édesítés:
  `.(Fej,Farok)` helyett a `[Fej|Farok]` kifejezést írjuk
- Kiterjesztés N darab „fej”-elemre, a skatulyázás kiküszöbölése:
  $[\texttt{Elem}_1|[\ldots|[\texttt{Elem}_N|\texttt{Farok}]\ldots]] \Longrightarrow [\texttt{Elem}_1,\ldots,\texttt{Elem}_N|\texttt{Farok}]$
- Ha a farok `[]`, a „`|[]`” jelsorozat elhagyható:
  $[\texttt{Elem}_1,\ldots,\texttt{Elem}_N|[]] \Longrightarrow [\texttt{Elem}_1,\ldots,\texttt{Elem}_N]$
- Egy `Kif` Prolog kifejezés nyílt végű lista, ha `Kif` változó,
  vagy `Kif = [_|Farok]` ahol `Farok` nyílt végű lista
  (azaz ha előbb-utóbb egy változó található a farokpozíción)

```
| ?- [1,2] = [X|Y].          ⟹   X = 1, Y = [2] ?
| ?- [1,2] = [X,Y].          ⟹   X = 1, Y = 2 ?
| ?- [1,2,3] = [X|Y].        ⟹   X = 1, Y = [2,3] ?
| ?- [1,2,3] = [X,Y].        ⟹   no
| ?- [1,2,3,4] = [X,Y|Z].    ⟹   X = 1, Y = 2, Z = [3,4] ?
| ?- L = [1|_], L = [_,2|_]. ⟹   L = [1,2|_A] ? % nyílt végű!
| ?- L = .(1,[2,3|[]]).      ⟹   L = [1,2,3] ?
| ?- L = [1,2|.(3,[])].      ⟹   L = [1,2,3] ?
```

Egy egyszerű listakezelő eljárás

```
% unalmas(Lista, X): Lista minden eleme = X
unalmas([], _X).
unalmas([H|T], X) :-
    H = X,
    unalmas(T, X).
```

```
% Ekvivalens eljárás
unalmas([], _).
unalmas([X|T], X) :-

    unalmas(T, X).
```

```
| ?- unalmas([1,2,3], _).            ⟹   no
| ?- unalmas([2,2,2], 1).            ⟹   no
| ?- unalmas([2,2,2], 2).            ⟹   yes
| ?- unalmas([2,2,2], X).            ⟹   X = 2 ? ; no
| ?- L=[_,_,_], unalmas(L, 3).       ⟹   L = [3,3,3] ? ; no
| ?- L=[_,_,_], unalmas(L, X).       ⟹   L = [X,X,X] ? ; no
| ?- length(L, 10), unalmas(L, X).  ⟹   L = [X,X,X,X,X,X,X,X,X,X] ? ; no
| ?- length(L, 10), unalmas(L, X), X = 5.
      ⟹   L = [5,5,5,5,5,5,5,5,5,5], X = 5 ? ; no
```

Lista-komprehenzió: megoldások listába gyűjtése

- A `findall(Gyűjtő, Cél, Lista)` eljárás
  - a `Cél` kifejezést eljáráshívásként értelmezi;
  - a `Cél` eljárást meghívja, és minden sikeres lefutása után a `Gyűjtő` adatstruktúra másolatát elmenti;
  - a `Cél` hívás keresési terének kimerítésekor a `Gyűjtő` változó összes elmentett másolatát (a keletkezésük sorrendjében) egy listába gyűjti és ezt egyesíti a `Lista` kimenő paraméterrel.
- Példák az eljárás használatára:

```
| ?- findall(NSz, nsz('Imre', NSz), NSzk).
        ⟹    NSzk = ['B. Gizella','C. Henrik','Sarolt','Géza'] ? ; no
| ?- findall(Sz, sz('István',Sz), Szulok).
        ⟹    Szulok = ['Sarolt','Géza'] ? ; no
| ?- findall(Gy, sz(Gy, 'István'), Gyerekek).
        ⟹    Gyerekek = ['Imre'] ? ; no
| ?- findall(Gy, sz(Gy, 'Imre'), Gyerekek).
        ⟹    Gyerekek = [] ? ; no
```

- Ha nincs megoldás, akkor (értelemszerűen) egy üres listát kapunk eredményül.

Aritmetikai beépített eljárások

- Egy aritmetikai kifejezés[43] (`AKif`) a BIP végrehajtásakor kötelezően:
  - tömör (ground) – behelyettesítetlen változót nem tartalmaz;
  - csak számokból és megengedett aritmetikai függvényekből áll
- A legfontosabb (2-arg.-ú) függvények: `+`, `-`, `*`, `/` (lebegőp. eredményt ad), `//` (egész-osztás, 0 felé kerekít), `rem` (maradék, `//` szerint)
- `X is AKif`: Az `AKif` aritmetikai kif. értékét egyesíti `X`-szel, pl.

```
| ?- X = 2, Y is X+1.                  ⟹  X = 2, Y = 3 ?; no
| ?- Y is X+1, X = 2.                  ⟹  ! Instantiation error
| ?- 3 is 2+1.                         ⟹  yes
| ?- 1+3 is 6-2.                       ⟹  no % a bal oldalt nem értékeli ki!
| ?- X = 1, Y is (X-27) rem (X+2).     ⟹  X = 1, Y = -2 ?; no
| ?- Kif = X/(X-1), X = 6, Y is Kif.   ⟹  Kif = 6/(6-1), X = 6, Y = 1.2 ?; no
```

- További aritmetikai BIP-ek: `AKif1 < AKif2`, `AKif1 > AKif2`, `AKif1 =< AKif2` (vigyázat: nem <=), `AKif1 >= AKif2`, `AKif1 =:= AKif2` (aritmetikailag egyenlő), `AKif1 =\= AKif2` (aritmetikailag nem-egyenlő) – ezek mindkét oldalt kiértékelik, és elvégzik a kért összehasonlítást:

```
| ?- 1+3 =:= 6-2.                      ⟹  yes
| ?- 1+1 =\= 6/3.                      ⟹  no
```

[43] pl. `https://sicstus.sics.se/sicstus/docs/latest/html/sicstus/ref_002dari_002daex.html`

Példa: faktoriális számítása Prologban

- Funkc. nyelven a faktoriális egy 1-argumentumú függvény: `F = fakt(N)`
- Prologban ennek egy kétargumentumú reláció felel meg: `fakt(N, F)`
- Konvenció: az utolsó argumentum(ok) a kimenő pararaméter(ek)
- Idézzük föl a faktoriális függvény Elixir megvalósítását:

```
def fakt0(0)              do 1               end
def fakt0(n) when n > 0   do n * fakt0(n-1)  end
```

- Írjuk át úgy, hogy a `fakt0` hívás különüljön el az egyéb számításoktól:

```
def fakt1(0)             do                                          1 end
def fakt1(n) when n > 0  do n1 = n-1; f1 = fakt1(n1); f = f1*n; f end
```

- Az „`F = fakt1(N)`” $\Longrightarrow$ `fakt(N, F)` transzformáció adja a Prolog kódot:

```
% fakt(N, F): F = N!.
fakt(0, 1).                    % 0! = 1.
fakt(N, F) :-                  % N! = F ha létezik olyan N1 és F1, hogy
    N > 0, N1 is N-1,          %    N > 0 és N1 = N-1 és
    fakt(N1, F1),              %    F1 = N1! és
    F is F1*N.                 %    F = F1*N.
```

```
| ?- fakt(5, F).  ⟹  F = 120 ? ; no
| ?- fakt(4, 24). ⟹  yes
| ?- fakt(0, F).  ⟹  F = 1 ? ; no
```

```
| ?- fakt(0, 2).  ⟹  no
| ?- fakt(N, 1).  ⟹  N = 0 ? ;
                      ! Inst.  err...
```

Aritmetika plusz lista-komprehenzió

Egy korábbi Elixir példa: gyűjtsük össze azokat a pitagoraszi számhármasokat, amelyek összege egy adott `N` számnál kisebb-egyenlő

```
:- use_module(library(between)). % SWI Prologban elhagyandó

% pitag(N, Hármasok):
% Hármasok = { p(A,B,C) | 1 =< A < B < C, A+B+C =< N, A*A + B*B = C*C }
pitag(N, Pk) :-
    findall(

      p(A,B,C),
      ( between(1, N, A),
        between(1, N, B), A < B,
        between(1, N, C), A+B+C =< N,
        A*A + B*B =:= C*C
      ),
      Pk).
```

```
% def pitag(n), do:
%
% (ls = 1..n
%
%  for a <- ls,
%      b <- ls, a < b,
%      c <- ls, a + b + c <= n,
%      a * a + b * b === c * c,
%        do: {a, b, c}
% )
```

```
| ?- pitag(12, Pk).      ⟹    Pk = [p(3,4,5)] ? ; no
| ?- pitag(36, Pk).
     ⟹    Pk = [p(3,4,5),p(5,12,13),p(6,8,10),p(9,12,15)] ? ; no
```

Programfejlesztési beépített predikátumok

- `consult(File)`: A `File` állományban levő programot beolvassa és értelmezendő alakban eltárolja. (`File = user` $\Rightarrow$ terminálról olvas.)
- `compile(File)` vagy `[File]`: mint `consult`, csak kompilált alakban tárol (gyorsabb kód, de egyes BIP-ek nem nyomkövethetők)
- `trace`, `notrace`: A (teljes) nyomkövetést be- ill. kikapcsolja.
- `listing` vagy `listing(Predikátum)`: Az értelmezendő alakban eltárolt összes ill. adott nevű predikátumokat kilistázza.
- `halt`: A Prolog rendszer befejezi működését.

```
> sicstus
SICStus 4.4.1 (x86_64-linux-glibc2.12) ...
| ?- consult(fakt).
% consulted /home/user/fakt.pl in module user, 10 msec 91776 bytes
yes
| ?- fakt(4, F).
F = 24 ? ;
no
| ?- listing(fakt).
(...)
yes
| ?- halt.
>
```

Adatstruktúrák Prologban – a bináris fák példája

- A bináris fa adatstruktúra
  - vagy egy csomópont (`node`), amelynek két részfája van (`left,right`)
  - vagy egy levél (`leaf`), amely egy egészt tartalmaz

Binárisfa-struktúra C-ben

```
enum treetype {Node, Leaf};
struct tree {
 enum treetype type;
 union {
   struct { struct tree *left;
            struct tree *right;
          } nd;
   struct { int value;
          } lf;
 } u;
};
```

A Prolog dinamikusan típusos nyelv – nincs szükség explicit típusdefinícióra

- Mercury típusleírás (komment)

```
% :- type tree --->
%         node(tree, tree)
%       | leaf(int).
```

- A típushoz tartozás ellenőrzése

```
% is_tree(T): T egy bináris fa
is_tree(leaf(V)) :- integer(V).
is_tree(node(Left,Right)) :-
    is_tree(Left),
    is_tree(Right).
```

Bináris fák összegzése

Egy bináris fa levélösszegének kiszámítása:

- egylevelű fa esetén a levélben tárolt egész
- csomópont esetén a két részfa levélösszegének összege

C nyelven

```
% S = tsum(T): T levélösszege S
int tsum(struct tree *tree)
{
 switch(tree->type) {
 case Leaf:
  return tree->u.lf.value;
 case Node:
  return tsum(tree->u.nd.left) +
         tsum(tree->u.nd.right);
   }
}
```

Prologban

```
% tree_sum(Tree, S): Σ Tree = S.
tree_sum(leaf(Value), Value).
tree_sum(node(Left,Right), S) :-
        tree_sum(Left, S1),
        tree_sum(Right, S2),
        S is S1+S2.
| ?- tree_sum(node(leaf(5),
                   node(leaf(3),
                        leaf(2))),S).
S = 10 ? ;
no
| ?- tree_sum(T, 3).
T = leaf(3) ? ;
! Inst. error in argument 2 of is/2
! goal:  3 is _73+_74
```

Tartalom

Predikátumok, klózok

- Példa:

```
% két klózból álló predikátum definíciója, funktora: tree_sum/2
tree_sum(leaf(Val), Val).          %                    1. klóz, tényáll.
tree_sum(node(Left,Right), S) :- %           fej    \
        tree_sum(Left, S1),        % cél     \       |
        tree_sum(Right, S2),       % cél     | törzs | 2. klóz, szabály
        S is S1+S2.                % cél     /       /
```

- Szintaxis:

⟨ Prolog program ⟩	::=	⟨ predikátum ⟩ . . .	
⟨ predikátum ⟩	::=	⟨ klóz ⟩ . . .	{azonos funktorú}
⟨ klóz ⟩	::=	⟨ tényállítás ⟩.␣ |	
		⟨ szabály ⟩.␣	{klóz funktora = fej funktora}
⟨ tényállítás ⟩	::=	⟨ fej ⟩	
⟨ szabály ⟩	::=	⟨ fej ⟩ :- ⟨ törzs ⟩	
⟨ törzs ⟩	::=	⟨ cél ⟩, . . .	
⟨ cél ⟩	::=	⟨ kifejezés ⟩	
⟨ fej ⟩	::=	⟨ kifejezés ⟩	

Prolog kifejezések

- Példa – egy klózfej mint kifejezés:

```
% tree_sum(node(Left,Right), S)    % összetett kif., funktora
% ________ _________________  _                      tree_sum/2
%      |          |            |
% struktúranév    \          argumentum, változó
%                  \- argumentum, összetett kif.
```

- Szintaxis:

⟨ kifejezés ⟩	::=	⟨ változó ⟩ |	{Nincs funktora}
		⟨ konstans ⟩ |	{Funktora: ⟨ konstans ⟩/0}
		⟨ összetett kif. ⟩ |	{Funktor: ⟨ struktúranév ⟩/⟨ arg.sz. ⟩}
		⟨ egyéb kifejezés ⟩	{Operátoros, lista, stb.}
⟨ konstans ⟩	::=	⟨ névkonstans ⟩ |	
		⟨ számkonstans ⟩	
⟨ számkonstans ⟩	::=	⟨ egész szám ⟩ |	
		⟨ lebegőp. szám ⟩	
⟨ összetett kif. ⟩	::=	⟨ struktúranév ⟩ (⟨ argumentum ⟩, . . .)	
⟨ struktúranév ⟩	::=	⟨ névkonstans ⟩	
⟨ argumentum ⟩	::=	⟨ kifejezés ⟩	

Lexikai elemek: példák és szintaxis

```
% változó:              Fakt FAKT _fakt X2 _2 _
% névkonstans:         fakt ≡ 'fakt' 'István' [] ; ',' += ** \= ≡ '\\='
% számkonstans:        0 -123 10.0 -12.1e8
% nem névkonstans:     !=, Istvan
% nem számkonstans:    1e8 1.e2
```

⟨ változó ⟩ ::= ⟨ nagybetű ⟩⟨ alfanum. jel ⟩... |
_ ⟨ alfanum. jel ⟩...

⟨ névkonstans ⟩ ::= ’⟨ idézett kar. ⟩...’ |
⟨ kisbetű ⟩⟨ alfanum. jel ⟩... |
⟨ tapadó jel ⟩... | ! | ; | [] | {}

⟨ egész szám ⟩ ::= {előjeles vagy előjeltelen számjegysorozat}

⟨ lebegőp.szám ⟩ ::= {belsejében tizedespontot tartalmazó
számjegysorozat esetleges exponenssel}

⟨ idézett kar. ⟩ ::= {tetszőleges nem ' és nem \ karakter} |
\ ⟨ escape szekvencia ⟩

⟨ alfanum. jel ⟩ ::= ⟨ kisbetű ⟩ | ⟨ nagybetű ⟩ | ⟨ számjegy ⟩ | _

⟨ tapadó jel ⟩ ::= + | - | * | / | \ | $ | ^ | < | > | = | ` | ~ | : | . | ? | @ | # | &

Prolog programok formázása

- Megjegyzések (comment)
  - A % százalékjeltől a sor végéig
  - A /* jelpártól a legközelebbi */ jelpárig.
- Formázó elemek (komment, szóköz, újsor, tabulátor stb.) szabadon használhatók, kivételek:
  - összetett kifejezésben a név után tilos formázó elemet tenni
  - prefix operátor (ld. később) és „(” között kötelező a formázó elem
  - klózt lezáró pont (.␣): önmagában álló pont (ha előtte tapadó jel áll, akkor a pont elé formázó elemet kell tenni), amit legalább egy formázó elem követ
- Programok javasolt formázása:
  - Az egy predikátumhoz tartozó klózok legyenek egymás mellett a programban, közéjük ne tegyünk üres sort
  - A predikátum elé tegyünk egy üres sort és egy fejkommentet:

```
% predikátumnév(A1, ..., An): A1, ..., An közötti
% összefüggést leíró kijelentő mondat.
```

  - A klózfejet írjuk a sor elejére, minden célt lehetőleg külön sorba, néhány szóközzel beljebb kezdve

Elixir és Prolog: néhány eltérés és hasonlóság

Elixir	**Prolog**
függvény, értéke tetsz. típusú	predikátum, azaz Boole-értékű függvény
arg. bemenő, a fv.érték kimenő	arg.-ok bemenők és kimenők is
egyetlen visszatérési érték	választási pontok, több megoldás lehet
külön ennes, lista típusok	a lista is összetett kifejezés
nincsenek felh. operátorok	felhasználói operátorok definiálhatók
Az = jobb oldalán tömör kif., bal oldalon mintakif.; őrfeltételekkel	az egyesítés szimmetrikus, mindkét oldalon minták

- Néhány hasonlóság:
  - az eljárás is klózokból áll, kiválasztás mintaillesztéssel, sorrendben, de míg Elixirben csak az első illeszkedő klózfej számít, Prologban az összes
  - változóhoz csak egyszer köthető érték
  - lista szintaxisa (de: Elixirben önálló típus), sztring (füzér), atom

Tartalom

Listák összefűzése – az `append/3` eljárás

- Elixir megoldás:

```
def append([],    b) do                    b     end
def append([x|a], b) do c = append(a,b); [x|c] end
```

- Írjuk át a kétargumentumú `append` függvényt egy `app0/3` Prolog eljárássá!

```
% app0(A, B, C): A és B listák összefűzése a C lista, C = A ⊕ B
app0([], B, Ret) :- Ret = B.
app0([X|A], B, Ret) :-
    app0(A, B, C), Ret = [X|C].
```

- Logikailag tiszta Prolog programokban a `Vált = Kif` alakú hívások elhagyhatók, ha `Vált` többi előfordulását `Kif`-re cseréljük.

```
app([], B, B).
app([X|A], B, [X|C]) :-
    app(A, B, C).
```

- Mindkét eljárásban a (max) futási idő arányos az 1. arg. hosszával
- Miért jobb az `app/3` mint az `app0/3`?
  - `app/3` jobbrekurzív, ciklussal ekvivalens (nem fogyaszt vermet)
  - `app([1,...,1000],[0],[2,...])` 1, `app0(...)` 1000 lépésben hiúsul meg.
  - `app/3` használható szétszedésre is (lásd később), míg `app0/3` nem.

Lista építése *elölről* – nyílt végű listákkal

- **Ismétlés:** egy `X` Prolog kifejezés nyílt végű lista, ha `X` változó, vagy `X = [_|Farok]` ahol `Farok` nyílt végű lista.

  `| ?- L = [1|_], L = [_,2|_].` $\Longrightarrow$ `L = [1,2|_A] ?`

- A beépített `append/3` azonos az `app/3`-mal:

```
append([], B, B).
append([X|A], B, [X|C]) :-
    append(A, B, C).
```

- Az `append` eljárás már az első redukciónál felépíti az eredmény fejét!
  - Példa-célsorozat: `append([1,2], [3,4,5], Ered), answer(Ered).`
  - Fej: `append([X|A], B, [X|C])`
  - Behelyettesítés: `X = 1, A = [2], B = [3,4,5], Ered = [1|C]`
  - Új célsorozat: `append([2], [3,4,5], C), answer([1|C]).`
    (`Ered` nyílt végű lista, farka még behelyettesítetlen.)

Lista építése *elölről* – a megvalósítás részletei

- A kimenő paraméter behelyettesítését explicitté tehetjük:

```
app1([], B, L) :-                                   % (a1)
    L = B.
app1([X|A], B, L) :-                                % (a2)
    L = [X|C],
    app1(A, B, C).
```

- Egy `app1/3` eljáráshívás redukciós lépései:

```
:- app1([1,2], [3,4,5], Ered), answer(Ered).                    % (cs0)
                                                  % + (a2) =>
:- Ered = [1|C1], app1([2], [3,4,5], C1), answer(Ered).         % (cs1)
                                                  % + BIP   =>
:- app1([2], [3,4,5], C1), answer([1|C1]).                      % (cs2)
                                                  % + (a2) =>
:- C1 = [2|C2], app1([], [3,4,5], C2), answer([1|C1]).          % (cs3)
                                                  % + BIP   =>
:- app1([], [3,4,5], C2), answer([1,2|C2]).                     % (cs4)
                                                  % + (a1) =>
:- C2 = [3,4,5], answer([1,2|C2]).                              % (cs5)
                                                  % + BIP   =>
:- answer([1,2,3,4,5]).                                         % (cs6)
```

Listák szétbontása az append/3 segítségével

```
% append(L1, L2, L3):
% Az L3 lista az L1 és L2
% listák elemeinek egymás
% után fűzésével áll elő.
append([], L, L).
append([X|L1], L2, [X|L3]) :-
    append(L1, L2, L3).

| ?- append(A, B, [1,2,3,4]).
A = [], B = [1,2,3,4] ? ;
A = [1], B = [2,3,4] ? ;
A = [1,2], B = [3,4] ? ;
A = [1,2,3], B = [4] ? ;
A = [1,2,3,4], B = [] ? ;
no
```



Nyílt végű listák az append változatokban

```
app0([], L, L).
app0([X|L1], L2, R) :-
    app0(L1, L2, L3), R = [X|L3].
```

```
append([], L, L).
append([X|L1], L2, [X|L3]) :-
    append(L1, L2, L3).
```

- Ha az 1. argumentum zárt végű (n hosszú), mindkét változat legfeljebb $n+1$ lépésben egyértelmű választ ad, amely lehet nyílt végű:
  `| ?- app0([1,2], L2, L3).` $\Longrightarrow$ `L3 = [1,2|L2] ? ; no`
- A 2. arg.-ot nem bontjuk szét $\Longrightarrow$ mindegy, hogy nyílt vagy zárt végű
- Ha a 3. argumentum zárt végű (n hosszú), akkor az `append` változat legfeljebb $n+1$ megoldást ad, max. $\sim 2n$ lépésben (ld. előző dia); tehát:
  - `append(L1, L2, L3)` keresési tere véges, ha `L1` vagy `L3` zárt
- Ha az 1. és a 3. arg. is nyílt, akkor a válaszhalmaz csak végtelen sok Prolog kifejezéssel fedhető le, pl.
  `_` $\oplus$ `[1]` $=$ `L` ($\equiv$`L` utolsó eleme `1`): `L = [1]; [_,1]; [_,_,1]; ...`
- `app0` szétszedésre nem jó, mert pl. `app0(L, [1], [])` $\Longrightarrow \infty$ ciklus, hiszen redukálva a 2. klózzal $\Longrightarrow$ `app0(L1, [1], L3), [] = [X|L3].`
- Az `append` eljárás jobbrekurzív, hála a logikai változó használatának

Variációk `append`-re – három lista összefűzése (kieg. anyag)

- `append(L1,L2,L3,L123)`: $(\texttt{L1} \oplus \texttt{L2}) \oplus \texttt{L3} = \texttt{L123}$

```
append(L1, L2, L3, L123) :-
        append(L1, L2, L12), append(L12, L3, L123).
```

- Lassú, pl.: `append([1,...,100],[1,2,3],[1], L)` 103 helyett 203 lépés!
- Szétszedésre nem alkalmas – végtelen választási pontot hoz létre
- Szétszedésre is alkalmas, hatékony változat

```
% L1 ⊕ (L2 ⊕ L3) = L123,
% ahol vagy L1 és L2, vagy L123 adott (zárt végű).
append(L1, L2, L3, L123) :-
        append(L1, L23, L123), append(L2, L3, L23).
```

  - `append(+,+,?,?)` esetén az első `append/3` hívás nyílt végű listát ad:

    `| ?- append([1,2], L23, L).` $\Longrightarrow$ `L = [1,2|L23] ?`

  - Az `L3` argumentum behelyettesítettsége (nyílt vagy zárt végű lista-e) nem számít.

Listák megfordítása

- Naív (négyzetes lépésszámú) megoldás

```
% nrev(L, R): R = L megfordítása.
nrev([], Ret) :- Ret = [].          % def nrev([])    do []  end
nrev([X|L], Ret) :-                 % def nrev([x|l]) do
    nrev(L, RL),                    %         rl = nrev(l);
    append(RL, [X], Ret).           %         append(rl,[x]) end
```

- Lineáris lépésszámú megoldás

```
% revapp(L0, R0, R): L0 megfordítását R0 elé fűzve kapjuk R-t.
revapp([], R0, R) :- R = R0.        % def revapp([], r0)     do r0 end
revapp([X|L0], R0, R) :-            % def revapp([x|l0], r0) do
    revapp(L0, [X|R0], R).          %         revapp(l0, [x|r0])   end

% reverse(R, L): Az R lista az L megfordítása.
reverse(R, L) :-  revapp(L, [], R).
```

- `revapp`-ban *R0*, *R* egy akkumulátorpár: eddigi ill. végeredmény
- A `lists` könyvtár tartalmazza a `reverse/2` eljárás definícióját, betöltése:

```
:- use_module(library(lists)).
```

Listák gyűjtése elölről és hátulról (kieg. anyag)

- Prolog

```
revapp([], L, L).
revapp([X|L0], L2, L3) :-
    revapp(L0, [X|L2], L3).
```

```
append([], L, L).
append([X|L1], L2, [X|L3]) :-
    append(L1, L2, L3).
```

- C++

```
struct lnk  { char elem;
              lnk *next;
              lnk(char e): elem(e) {}   };

typedef lnk *list;
list revapp(list L1, list L2)
{ list l = L2;
  for (list p=L1; p; p=p->next)
  { list newl = new lnk(p->elem);
    newl->next = l; l = newl;
  }
  return l;
}
```

```
list append(list L1, list L2)
{ list L3, *lp = &L3;
  for (list p=L1; p; p=p->next)
  { list newl = new lnk(p->elem);
    *lp = newl; lp = &newl->next;
  }
  *lp = L2;  return L3;
}
```

Keresés listában – a `member/2` beépített eljárás

- `member(E, L)`: `E` az `L` lista eleme

```
member(Elem, [Elem|_]).
member(Elem, [_|Farok]) :-
    member(Elem, Farok).
```

- Eldöntendő (igen-nem) kérdés:

```
| ?- member(2, [1,2,3,2]).          ⟹   yes                          DE
| ?- member(2, [1,2,3,2]), R=yes.   ⟹   R=yes ? ; R=yes ? ; no
```

- Lista elemeinek felsorolása:

```
| ?- member(X, [1,2,3]).            ⟹   X = 1 ? ; X = 2 ? ; X = 3 ? ; no
| ?- member(X, [1,2,1]).            ⟹   X = 1 ? ; X = 2 ? ; X = 1 ? ; no
```

- Listák közös elemeinek felsorolása – az előző két hivásformát kombinálja:

```
| ?- member(X, [1,2,3]),
     member(X, [5,4,3,2,3]).        ⟹   X = 2 ? ; X = 3 ? ; X = 3 ? ; no
```

- Egy értéket egy (nyílt végű) lista elemévé tesz, végtelen választás!

```
| ?- member(1, L).                  ⟹   L = [1|_A] ? ; L = [_A,1|_B] ? ;
                                        L = [_A,_B,1|_C] ? ; ...
```

- A `member/2` keresési tere véges, ha 2. argumentuma zárt végű lista.

A `member/2` predikátum általánosítása: `select/3`

- `select(E, Lista, M)`: `E` elemet `Lista`ból pont egyszer elhagyva marad `M`.

```
select(E, [E|Marad], Marad).     % Elhagyjuk a fejet, marad a farok.
select(E, [X|Farok], [X|M]) :-   % Marad a fej,
        select(E, Farok, M).     %    a farokból hagyunk el elemet.
```

- Felhasználási lehetőségek:

```
| ?- select(1, [2,1,3,1], L).    % Adott elem elhagyása
    ⟹      L = [2,3,1] ? ; L = [2,1,3] ? ; no
| ?- select(X, [1,2,3], L).      % Akármelyik elem elhagyása
    ⟹      L=[2,3], X=1 ? ;  L=[1,3], X=2 ? ;   L=[1,2], X=3 ? ; no
| ?- select(3, L, [1,2]).        % Adott elem beszúrása!
    ⟹      L = [3,1,2] ? ; L = [1,3,2] ? ; L = [1,2,3] ? ; no
| ?- select(3, [2|L], [1,2,7,3,2,1,8,9,4]).
     % Beszúrható-e 3 az [1,...]-ba úgy, hogy [2,...]-t kapjunk?
    ⟹      no
| ?- select(1, [X,2,X,3], L).
    ⟹      L = [2,1,3], X = 1 ? ; L = [1,2,3], X = 1 ? ; no
```

- A `select/3` eljárás keresési tere véges, ha vagy a 2., vagy a 3. argumentuma zárt végű lista (a `lists` könyvtár tartalmazza az eljárást)

Listák permutációja (kieg. anyag)

- `% perm(+Lista, ?Perm)`: `Lista` permutációja a `Perm` lista.

```
perm0([], []).
perm0([Elso|Lista], Perm) :-
        perm0(Lista, Perm0),          % permutáljuk a bemenet farkát
        select(Elso, Perm, Perm0).    % ebbe beszúrjuk a bemenet fejét
```

- Felhasználási példák:

```
| ?- perm0([1,2], L).
    ⟹     L = [1,2] ? ; L = [2,1] ? ; no

| ?- perm0([a,b,c], L).
    ⟹     L = [a,b,c] ? ; L = [b,a,c] ? ; L = [b,c,a] ? ;
           L = [a,c,b] ? ; L = [c,a,b] ? ; L = [c,b,a] ? ; no

| ?- perm0(L, [1,2]).
    ⟹      L = [1,2] ? ;  végtelen keresési tér
```

- Ha `perm0/2`-ben az első argumentum változó, akkor a rekurzív hívás mindkét argumentuma változó lesz ⇒végtelen sok választás
- A `lists` könyvtárban van egy kétirányban működő `permutation/2` eljárás

Tartalom

Függvények és eljárások egymásba skatulyázása

- A deklaratív nyelvekben a rekurzió váltja ki a ciklust, így gyakran előfordulnak egymásba **skatulyázott** függvény- ill. eljáráshívások.
- Tekintsük a faktoriális Elixir definícióját!

```
% fac(N) = N faktoriálisa.
def fac(0), do: 1
def fac(n), do: n * fac(n-1)
```

- A `fac(4)` függvényhívás végrehajtásakor pl. az alábbi állapotokat kapjuk:



- A függvényhívásokba való be- és kilépés nyomon követése:

```
Call fac(4)
Call fac(3)
...
Call fac(0)
Exit fac(0) = 1
...
Exit fac(3) = 6
Exit fac(4) = 24
```

A `Call` nyomkövetési információ egy fenti doboz *létrehozásához* kapcsolható, míg az `Exit` a doboz kiértékelésének *befejezéséhez*.

Prolog nyomkövetés eljárás-doboz modellel

- A Prolog doboz alapú nyomkövetésében is az eljárások be- és kilépési pontjain (ún. kapukon, angolul *port*) való áthaladásról kapunk információt:
  - **Call port** (hívás kapu) – belépés az eljárásba, doboz létrehozása
  - **Exit port** (kilépés kapu) – sikeres lefutás, esetleg doboz törlése
  - **Fail port** (meghiúsulás kapu) – sikertelen lefutás, doboz törlése
  - **Redo port** (újra kapu) – új megoldás kérése
- A Prolog eljárás-végrehajtás két fázisa
  - előre menő: egymásba **skatulyázott eljárás-be** és **-ki**lépések
  - visszafelé menő: **új megoldás** kérése egy már lefutott eljárástól
- Prolog végrehajtás objektum-orientált szemléletben (eljárás $\Rightarrow$ objektum):
  - eljárás meghívása (hívás kapu): objektum létrehozása
  - sikeres lefutás (kilépés kapu): változóbehelyettesítések visszaadása
  - sikertelen lefutás (meghiúsulás kapu): meghiúsulás jelzése
  - új megoldás kérése (újra kapu): további választási pont(ok) bejárása

Eljárás-doboz modell – grafikus szemléltetés

Egy egyszerű példaprogram, hívása `| ?- p(X).`

```
q(2).  q(4).  q(7).          p(X) :- q(X), X > 3.
```



Előre: `Call p(X); Call q(X); Exit q(2); Call 2>3; Fail 2>3`

Vissza: `Redo q(2);`

Előre: `Exit q(4); Call 4>3; Exit 4>3; Exit p(4);` siker `X = 4 ? ;`

Vissza: `Redo p(4); Redo 4>3; Fail 4>3; Redo q(4);`

Előre: `Exit q(7); Call 7>3; Exit 7>3; Exit p(7);` siker `X = 7 ? ;`

Vissza: `Redo p(7); Redo 7>3; Fail 7>3; ...; Fail p(X);` meghiúsulás `no`

Egy egyszerű nyomkövetési példa (SICStus Prolog)

- SICStusban `?...Exit` jelzi, hogy van választási pont a lefutott eljárásban
- Ha nincs `?` az Exit kapunál, akkor a doboz törlődik (lásd a szaggatott piros téglalapot az `X > 3` hívás körül)

```
q(2).
q(4).
q(7).

p(X) :- q(X), X > 3.
```




```
% Sorszám      Mélység
| ?- consult(pq0), trace, p(X).     % compile esetén a > /2 hívásokat nem látjuk
        1      1 Call: p(_463) ?
        2      2 Call: q(_463) ?
?       2      2 Exit: q(2) ?      % ? ≡ maradt választási pont q-ban
        3      2 Call: 2>3 ?
        3      2 Fail: 2>3 ?
        2      2 Redo: q(2) ?
?       2      2 Exit: q(4) ?
        4      2 Call: 4>3 ?
        4      2 Exit: 4>3 ?       % nincs ? ⇒ a doboz törlődik, (ld. a szaggatott piros téglalapot)
?       1      1 Exit: p(4) ?
X = 4 ? ;
```

Egy egyszerű nyomkövetési példa (folyt.)

```
q(2).
q(4).
q(7).

p(X) :- q(X), X > 3.
```




```
| ?- consult(pq0), trace, p(X).
(...)
        4      2 Exit: 4>3 ?        % nincs ? ⇒ a doboz törlődik (*)
?       1      1 Exit: p(4) ?
X = 4 ? ;
        1      1 Redo: p(4) ?
                                    % (*) miatt nem látjuk a Redo-Fail kapukat a 4>3 hívásra
        2      2 Redo: q(4) ?
        2      2 Exit: q(7) ?       % nincs ? ⇒ a doboz törlődik
        5      2 Call: 7>3 ?
        5      2 Exit: 7>3 ?        % nincs ? ⇒ a doboz törlődik
        1      1 Exit: p(7) ?       % nincs ? ⇒ a doboz törlődik
X = 7 ? ;
no
```

Eljárás-doboz: több klózból álló eljárás

```
ős(X,Y) :- sz(X,Z), ős(Z,Y). % X őse Y ha X szülője Z és Z őse Y (a szülő őse ős)
ős(X,Y) :- sz(X,Y).          % X őse Y ha X szülője Y           (a szülő ős)

sz(a,b).   sz(b,c).   sz(b,d).
```



Eljárás-doboz modell – „kapcsolási” alapelvek

- A feladat: „szülő” eljárásdoboz és a „belső” eljárások dobozainak összekapcsolása
- Előfeldolgozás: érjük el, hogy a klózfejekben csak változók legyenek, ehhez a fej-egyesítéseket alakítsuk hívásokká, pl.
  `fakt(0,1).` $\Rightarrow$ `fakt(X,Y) :- X=0, Y=1.`
- Előre menő végrehajtás (balról-jobbra menő nyilak):
  - A szülő Call kapuját az 1. klóz első hívásának Call kapujára kötjük.
  - Egy belső eljárás Exit kapuját
    - a következő hívás Call kapujára, vagy,
    - ha nincs következő hívás, akkor a szülő Exit kapujára kötjük
- Visszafelé menő végrehajtás (jobbról-balra menő nyilak):
  - Egy belső eljárás Fail kapuját
    - az előző hívás Redo kapujára, vagy, ha nincs előző hívás, akkor
    - a következő klóz első hívásának Call kapujára, vagy
    - ha nincs következő klóz, akkor a szülő Fail kapujára kötjük
  - A szülő Redo kapuját mindegyik klóz utolsó hívásának Redo kapujára kötjük
    - mindig abba a klózra térünk vissza, amelyben legutoljára voltunk

Nyomkövetés – legfontosabb parancsok (SICStus + SWI)

- Beépített eljárások
  - `trace`, `debug` – a `c`, `l` paranccsal indítja a nyomkövetést
  - `notrace`, `nodebug` – kikapcsolja a nyomkövetést
  - `spy(P)`, `nospy(P)`, `nospyall` – töréspont be/ki a `P` eljárásra, $\forall$ ki.
- Alapvető nyomkövetési parancsok (SICStus: `<RET>`-tel kell lezárni)
  - `h` (help) – parancsok listázása
  - `c` (creep) vagy csak `<RET>` – lassú futás (minden kapunál megáll)
  - `l` (leap) – csak töréspontnál áll meg
  - `+` ill. `-` – töréspont be/ki a kurrens eljárásra
  - `s` (skip) – eljárástörzs átlépése (Call/Redo $\Rightarrow$ Exit/Fail)
  - `w` (write) – teljes mélységű kiíratás
  - `o` (out) SICStus, `u` (up) SWI – kilépés az eljárástörzsből
  - `r` (retry) – újrakezdi a kurrens hívás végrehajtását
- Információ-megjelenítő és egyéb parancsok
  - `g` (goals) – a kurrens hívást tartalmazó célok kiíratása
  - `b` (break) – új, beágyazott Prolog interakciós szint létrehozása
  - `n` (notrace) – nyomkövető kikapcsolása
  - `a` (abort) – a kurrens futás abbahagyása